

Appendix A - Ismene Grammar

The following describes the Ismene grammar. A *word* represents a string of non-whitespace alphanumeric characters. A *string* is a string of alphanumeric characters (i.e., may contain newline and whitespace characters).

```
<policy> := <statements>
<statements> := <statement> ";" ["," <statements>]
<statement> := <attribute> | <prov_clause> | <action_clause> | <assertion>

<attribute> := <identifier> "==" "<" <value> ">" |
               <identifier> "==" "<" <value list> ">"
<value list> := "{" <value> "}" ["," <value list>]
<identifier> := word
<value> := string

<prov_clause> := <tag> ":" [<conditionals>] "::" <consequences>
<tag> := <identifier>
<conditionals> := <var> "=" <value> ["," <conditionals>] |
                  <predicate> "(" [<args> "]" ["," <conditionals>]
<var> := "$" <identifier>
<predicate> := <identifier>
<args> := <var> | <identifier> ["," <args>]
<consequences> := <pick> | <config> | <tag> ["," <consequences>]
<pick> := "pick" "(" <config> "," <config> ["," <configs> "]"
<configs> := <config> ["," <configs>]
<config> := "config" "(" [ <cfgstmts> ] ")"
<cfgstmts> := <mechanism> [ "(" <params> ")" ] [, <cfgstmts>]
<mechanism> := <identifier>
<params> := <identifier> "=" <value> ["," <params> ]

<action_clause> := <tag> ":" [<action_conditionals>]
                  "::" "accept" [, "reconfig"]
<action_conditionals> := <action_condition> ["," <action_conditionals>]
<action_condition> := <conditionals> | <credential> | <config> | <pick>
<credential> := Credential "(" <bind var> "," <cred_args> ")"
<bind var> := "&" <identifier>
<cred_args> := <identifier> "=" <value> ["," <cred_args>] |
               <identifier> "=" <var> ["," <cred_args>] |
               <identifier> "=" <bind var> ["," <cred_args>]

<assertion> := "assert" ":" [<assert_conds>] "::" <configs>;
<assert_conds> := <condition> | <config> ["," <assert_conds>]
```